A model to study the impact of government-imposed social distancing on COVID-19 epidemic in Portugal

Clearing memory

In[3548]:=
```
ClearSystemCache[]
ClearAll["Global`*"]
Clear["Subscript"]
Clear["Superscript"]
Clear["Subsuperscript"]
```

O estado de emergência: 18/03/2020, 17th day after 02/03/2020 (date of first confirmed cases in Portugal)
O dia de hoje: 01/04/2020, 31th day after 02/03/2020 (date of first confirmed cases in Portugal)

In[3553]:=
```
Emergencia = 17 / 365;
Hoje = 31 / 365;
```

Model equations

In[3555]:=
```
eq[Reduction_][1] :=
 S'[t] == -S[t] ((β (σ IM[t] + IS[t])) If[t ≤ Emergencia, 1, If[t ≤ (Hoje - 6 / 365),
          7 / 13.64, Reduction]]) / (NN[t] - IQ[t]))
eq[Reduction_][2] := EE'[t] == S[t] ((β (σ IM[t] + IS[t])) If[t ≤ Emergencia, 1,
          If[t ≤ (Hoje - 6 / 365), 7 / 13.64, Reduction]]) / (NN[t] - IQ[t])) - α EE[t]
eq[Reduction_][3] := IM'[t] == p α EE[t] - γ1 IM[t]
eq[Reduction_][4] := IS'[t] == (1 - p) α EE[t] - ν IS[t]
eq[Reduction_][5] := IQ'[t] == ν IS[t] - γ2 IQ[t] - η IQ[t]
eq[Reduction_][6] := RM'[t] == γ1 IM[t]
eq[Reduction_][7] := RS'[t] == γ2 IQ[t]
eq[Reduction_][8] := DD'[t] == η IQ[t]
```

Numer of variables in the model (including deceased individuals)

In[3563]:=
```
numvar = 8
eqs[Reduction_] := Table[eq[Reduction][i], {i, 1, numvar}]
lhs[Reduction_] := eqs[Reduction][[All, 1]];
rhs[Reduction_] := eqs[Reduction][[All, 2]];
TableForm[eqs[Reduction]]
```

Out[3563]= 8

Out[3567]//TableForm=

$$S'[t] == -\frac{\beta\,\text{If}\left[t\le\frac{17}{365},1,\text{If}\left[t\le\text{Hoje}-\frac{6}{365},\frac{7}{13.64},\text{Reduction}\right]\right]\,(\sigma\,\text{IM}[t]+\text{IS}[t])\,\text{S}[t]}{-\text{IQ}[t]+\text{NN}[t]}$$

$$EE'[t] == -\alpha\,EE[t] + \frac{\beta\,\text{If}\left[t\le\frac{17}{365},1,\text{If}\left[t\le\text{Hoje}-\frac{6}{365},\frac{7}{13.64},\text{Reduction}\right]\right]\,(\sigma\,\text{IM}[t]+\text{IS}[t])\,\text{S}[t]}{-\text{IQ}[t]+\text{NN}[t]}$$

$$IM'[t] == p\,\alpha\,EE[t] - IM[t]\,\gamma_1$$

$$IS'[t] == (1-p)\,\alpha\,EE[t] - \nu\,IS[t]$$

$$IQ'[t] == -\eta\,IQ[t] + \nu\,IS[t] - IQ[t]\,\gamma_2$$

$$RM'[t] == IM[t]\,\gamma_1$$

$$RS'[t] == IQ[t]\,\gamma_2$$

$$DD'[t] == \eta\,IQ[t]$$

Model variables

In[3568]:=
```
vars = {S[t], EE[t], IM[t], IS[t], IQ[t], RM[t], RS[t], DD[t]}
```

Out[3568]= {S[t], EE[t], IM[t], IS[t], IQ[t], RM[t], RS[t], DD[t]}

Total population size N(t) is not constant due to disease-related mortality

In[3569]:=
```
NN[t] = S[t] + EE[t] + IM[t] + IS[t] + IQ[t] + RM[t] + RS[t]
```

Out[3569]= EE[t] + IM[t] + IQ[t] + IS[t] + RM[t] + RS[t] + S[t]

Epidemiological parameters of the model

Average contact rate (unique persons), 1/year

In[3570]:=
```
AverageContactRate = c → 13.74 × 365
```

Out[3570]= c → 5015.1

Relative infectivity of mildly infected

In[3571]:=
```
RelativeInfectivity = σ → 0.5
```

Out[3571]= σ → 0.5

1/latent period, 1/year

In[3572]:= **RateInfectiousnessOnset = α → 365 / 4**

Out[3572]= $\alpha \to \frac{365}{4}$

Proportion of mildly infected

In[3573]:= **ProportionMildSymptoms = p → 0.8**

Out[3573]= p → 0.8

1/recovery period of mildly infected, 1/year

In[3574]:= **RecoveryRateMildSymptoms = γ1 → 365 / 7**

Out[3574]= $\gamma_1 \to \frac{365}{7}$

1/delay from onset of infectiousness to diagnosis for individuals with severe symptoms, 1/year

In[3575]:= **DiagnosisRate = ν → 365 / 5**

Out[3575]= ν → 73

1/delay from diagnosis to recovery, 1/year

In[3576]:= **RecoveryRateSevereSymptoms = γ2 → 365 / 14**

Out[3576]= $\gamma_2 \to \frac{365}{14}$

Case fatality rate

In[3577]:= **FatalityRate = f → 187 / 8251**

Out[3577]= $f \to \frac{187}{8251}$

Disease-associated death rate, 1/year

In[3578]:= **DeathRateDiagnosed = η → γ2 f / (1 - f) /. {RecoveryRateSevereSymptoms, FatalityRate}**

Out[3578]= $\eta \to \frac{68\,255}{112\,896}$

Basic reproduction number

In[3579]:= **BasicReproductionNumber = R0 → 6.45**

Out[3579]= $R_0 \to 6.45$

Probability of transmission per contact with infectious with severe symptoms

In[3580]:=
```
TransmissionProbability = Solve[R0 == p β σ / γ1 + (1 - p) β / ν /. β → c ϵ, ϵ][[1, 1]] /.
  {ProportionMildSymptoms, AverageContactRate, RelativeInfectivity,
   RecoveryRateMildSymptoms, DiagnosisRate, BasicReproductionNumber}
```

Out[3580]= $\epsilon \to 0.123535$

Transmission rate of infection via contact with infectious with severe symptoms, 1/year

In[3581]:=
```
TransmissionRate = β → c ϵ /. {AverageContactRate, TransmissionProbability}
```

Out[3581]= $\beta \to 619.539$

Parameters of the model

In[3582]:=
```
Parameters := {AverageContactRate, RelativeInfectivity, RateInfectiousnessOnset,
  ProportionMildSymptoms, RecoveryRateMildSymptoms, DiagnosisRate,
  RecoveryRateSevereSymptoms, FatalityRate, DeathRateDiagnosed,
  BasicReproductionNumber, TransmissionProbability, TransmissionRate}
```

Critical value for contact rate reduction where Reff < 1

In[3583]:=
```
(1 - 1 / (p β σ / γ1 + (1 - p) β / ν) /. Parameters) 100
```

Out[3583]= 84.4961

Solving differential equations

Start time, year
The first day of the simulation (initial condition) corresponds to 02/03/2020, when 2 COVID-19 cases were confirmed

In[3584]:=
```
tstart = 1 / 365
```

Out[3584]= $\frac{1}{365}$

End time, year
The simulation runs for 365 days

In[3585]:=
```
tend = 365 / 365;
```

Total population size at the beginning of an outbreak

In[3586]:=
```
Ntot = 10.2 × 10^6
```

Out[3586]= 1.02×10^7

Initial number of infected individuals

In[3587]:=
```
InfInit = 12
```

Out[3587]= 12

Initial conditions

In[3588]:=
```
ics = Table[ic[i], {i, 1, numvar}];

ic[1] = (Ntot - 13.74 InfInit - InfInit - 2) == vars[[1]] /. {t → tstart}
ic[2] = 13.74 InfInit == vars[[2]] /. {t → tstart}
ic[3] = 0.8 InfInit == vars[[3]] /. {t → tstart}
ic[4] = 0.2 InfInit == vars[[4]] /. {t → tstart}
ic[5] = 2 == vars[[5]] /. {t → tstart}
ic[6] = 0 == vars[[6]] /. {t → tstart}
ic[7] = 0 == vars[[7]] /. {t → tstart}
ic[8] = 0 == vars[[8]] /. {t → tstart}
```

Out[3589]= $1.01998 \times 10^7 == S\left[\frac{1}{365}\right]$

Out[3590]= $164.88 == EE\left[\frac{1}{365}\right]$

Out[3591]= $9.6 == IM\left[\frac{1}{365}\right]$

Out[3592]= $2.4 == IS\left[\frac{1}{365}\right]$

Out[3593]= $2 == IQ\left[\frac{1}{365}\right]$

Out[3594]= $0 == RM\left[\frac{1}{365}\right]$

Out[3595]= $0 == RS\left[\frac{1}{365}\right]$

Out[3596]= $0 == DD\left[\frac{1}{365}\right]$

Solution

In[3597]:=
```
solution[Reduction_, Parameters_] :=
  NDSolve[Join[eqs[Reduction], ics] /. Parameters, vars, {t, tstart, tend}];
```

Computing peak number of confirmed cases

In[3598]:=
```
Peak[Reduction_, Parameters_] :=
 Max[Flatten[Table[Evaluate[IQ[t] /. First@solution[Reduction, Parameters]],
    {t, tstart, tend, 1 / 365}]]]
PeakBaseline = Peak[1, Parameters]
Peak75 = Peak[0.25, Parameters]
Peak90 = Peak[0.1, Parameters]
Peak90 0.3
Peak85 = Peak[0.15, Parameters]
```

Out[3599]= 888 954.

Out[3600]= 220 858.

Out[3601]= 9093.3

Out[3602]= 2727.99

Out[3603]= 12 773.

Computing time until the peak number of confirmed cases since 02/03/2020 (days)

In[3604]:=
```
PeakTiming[Reduction_, Parameters_] := Ordering[
    Flatten[Table[Evaluate[IQ[t] /. First@solution[Reduction, Parameters]],
      {t, tstart, tend, 1 / 365}]], -1][[1]]
PeakTimingBaseline = PeakTiming[1, Parameters]
PeakTiming75 = PeakTiming[0.25, Parameters]
PeakTiming90 = PeakTiming[0.1, Parameters]
PeakTiming85 = PeakTiming[0.15, Parameters]
```

Out[3605]= 55

Out[3606]= 117

Out[3607]= 40

Out[3608]= 60

Data for confirmed cases - deaths - recoveries
Data is split in 02/03/2020-18/03/2020 and 19/03/2020-now
Source https://covid19.min-saude.pt/ponto-de-situacao-atual-em-portugal/

In[3609]:=
```
DataBefore =  {{1 / 365, (2 - 0)}, {2 / 365, (4 - 0)}, {3 / 365, (6 - 0)},
    {4 / 365, (9 - 0)}, {5 / 365, (13 - 0)}, {6 / 365, (21 - 0)}, {7 / 365, (30 - 0)},
    {8 / 365, (39 - 0)}, {9 / 365, (41 - 0)}, {10 / 365, (59 - 0)}, {11 / 365, (78 - 0)},
    {12 / 365, (112 - 0)}, {13 / 365, (169 - 1)}, {14 / 365, (245 - 2)},
    {15 / 365, (331 - 3)}, {16 / 365, (448 - 3 - 1)}, {17 / 365, (642 - 3 - 2)}};
DataAfter = {{18 / 365, (785 - 3 - 3)}, {19 / 365, (1020 - 5 - 6)},
    {20 / 365, (1280 - 5 - 12)}, {21 / 365, (1600 - 5 - 14)}, {22 / 365, (2060 - 14 - 23)},
    {23 / 365, (2362 - 30 - 22)}, {24 / 365, (2995 - 22 - 43)}, {25 / 365, (3544 - 43 - 60)},
    {26 / 365, (4268 - 43 - 76)}, {27 / 365, (5170 - 43 - 100)},
    {28 / 365, (5962 - 43 - 119)}, {29 / 365, (6408 - 43 - 140)},
    {30 / 355, (7443 - 43 - 160)}, {31 / 355, (8251 - 43 - 187)}};
```

Plotting numero de casos confirmados

In[3747]:=
```
ymax = Last[DataAfter][[2]] 1.05;
ymin = -200;
tmax = Hoje + 2 / 365;
tmin = tstart - 1 / 365;
LabelBaseline = "26/04";
Label75 = "27/06";
Label90 = "11/04";
```

```
Label85 = "01/05";


fig1 =
 Table[Show[Plot[{Evaluate[IQ[t] /. solution[1, Parameters]]}, {t, tstart, tend},
     AspectRatio → 0.4, ImageSize → 800, PlotRangePadding → None, Filling → Axis,
     PlotRange → {{tmin, tmax}, {ymin, ymax}}, AxesOrigin → {0, 0}, Frame →
      {{True, False}, {True, False}}, FrameStyle → Directive[Black, 17, Bold],
     PlotStyle → {Thickness[0.01], RGBColor[217 / 255, 0, 0]},
     FillingStyle -> Directive[Opacity[0.125]],
     FrameLabel → {{"Número de casos", None}, {None, None}},
     PlotLabel → Style["Evolução diária do número de casos confirmados ativos",
       17, Black, Bold],
     FrameTicks → {{Automatic, None}, {{{1 / 365, "02/03"}, {5 / 365, "06/03"},
         {10 / 365, "11/03"}, {15 / 365, "16/03"}, {20 / 365, "21/03"},
         {25 / 365, "26/03"}, {30 / 365, "31/03"}}, None}}],
    ListPlot[{DataBefore, DataAfter}],
    Graphics[{Black, Dashed, Thick,
      Line[{{Emergencia, ymin}, {Emergencia, ymax}}]}],
    Graphics[{Red, Line[{{PeakTimingBaseline / 365, ymin},
        {PeakTimingBaseline / 365, PeakBaseline}}]}], Graphics[{Red,
      Line[{{tmin, PeakBaseline}, {PeakTimingBaseline / 365, PeakBaseline}}]}],
    Graphics[Text[StyleForm["O estado\nde emergência", FontSize → 17,
       Bold, FontColor → Black], {20 / 365, 400}]]], {i, 1, Length[vars]}][[1]]

ymax = Last[DataAfter][[2]] 1.2;
ymin = 1;
tmax = Hoje + 2 / 365;
tmin = tstart - 1 / 365;

fig2 =
 Table[Show[LogPlot[{Evaluate[IQ[t] /. solution[1, Parameters]]}, {t, tstart, tend},
     AspectRatio → 0.4, ImageSize → 800, PlotRangePadding → None, Filling → Axis,
     PlotRange → {{tmin, tmax}, {ymin, ymax}}, AxesOrigin → {0, 0}, Frame →
      {{True, False}, {True, False}}, FrameStyle → Directive[Black, 17, Bold],
     PlotStyle → {Thickness[0.01], RGBColor[217 / 255, 0, 0]},
     FillingStyle -> Directive[Opacity[0.125]],
     FrameLabel → {{"Número de casos", None}, {None, None}},
     PlotLabel → Style["Evolução diária do número de casos confirmados ativos",
       17, Black, Bold],
     FrameTicks → {{Automatic, None}, {{{1 / 365, "02/03"}, {5 / 365, "06/03"},
         {10 / 365, "11/03"}, {15 / 365, "16/03"}, {20 / 365, "21/03"},
         {25 / 365, "26/03"}, {30 / 365, "31/03"}}, None}}],
    ListLogPlot[{DataBefore, DataAfter}],
    Graphics[{Black, Dashed, Thick,
      Line[{{Emergencia, 0}, {Emergencia, ymax}}]}],
    Graphics[{Red, Line[{{PeakTimingBaseline / 365, ymin},
        {PeakTimingBaseline / 365, PeakBaseline}}]}], Graphics[{Red,
      Line[{{tmin, PeakBaseline}, {PeakTimingBaseline / 365, PeakBaseline}}]}],
    Graphics[Text[StyleForm["O estado\nde emergência", FontSize → 17,
       Bold, FontColor → Black], {20 / 365, 400}]]], {i, 1, Length[vars]}][[1]]

ymax = PeakBaseline 1.05;
ymin = -25 000;
tmax = 240 / 365;
tmin = tstart - 7 / 365;

fig3 = Table[Show[Plot[{Evaluate[IQ[t] /. solution[1, Parameters]],
      Evaluate[IQ[t] /. solution[0.25, Parameters]]}, {t, tstart, tend},
```

```
      AspectRatio → 0.4, ImageSize → 800, PlotRangePadding → None,
      Filling → Axis, PlotRange → {{tmin, tmax}, {ymin, ymax}},
      AxesOrigin → {0, 0}, Frame → {{True, False}, {True, False}},
      FrameStyle → Directive[Black, 17, Bold],
      PlotStyle → {{Thickness[0.01], RGBColor[217 / 255, 0, 0]},
        {Thickness[0.01], RGBColor[26 / 255, 94 / 255, 214 / 255]}},
      FillingStyle -> Directive[Opacity[0.125]],
      FrameLabel → {{"Número de casos", None}, {None, None}}, PlotLegends →
       Placed[{Table[Style[Row[{label}], Black, 17, "Text", Bold], {label,
            {"Sem distanciamento social", "Redução de contactos em 75%"}}]},
        {Scaled[{0, 0.75}], {-1.25, 0.6}}], PlotLabel →
       Style["Evolução diária do número de casos confirmados ativos",
        17, Black, Bold],
      FrameTicks → {{Automatic, None}, {{{1 / 365, "02/03"},
          {PeakTimingBaseline / 365, LabelBaseline},
          {Emergencia, "18/03"}, {PeakTiming75 / 365, Label75}}, None}}],
     ListPlot[{DataBefore, DataAfter}],
     Graphics[{Black, Dashed, Thick,
       Line[{{Emergencia, ymin}, {Emergencia, ymax}}]}],
     Graphics[{RGBColor[217 / 255, 0, 0], Line[{{PeakTimingBaseline / 365, ymin},
         {PeakTimingBaseline / 365, PeakBaseline}}]}],
     Graphics[{RGBColor[217 / 255, 0, 0], Line[{{tmin, PeakBaseline},
         {PeakTimingBaseline / 365, PeakBaseline}}]}],
     Graphics[{RGBColor[26 / 255, 94 / 255, 214 / 255],
       Line[{{PeakTiming75 / 365, ymin}, {PeakTiming75 / 365, Peak75}}]}],
     Graphics[{RGBColor[26 / 255, 94 / 255, 214 / 255], Line[{{tmin, Peak75},
         {PeakTiming75 / 365, Peak75}}]}]], {i, 1, Length[vars]}][[1]]

ymax = Peak90 1.05;
ymin = -200;
tmax = 140 / 365;
tmin = t_start - 4 / 365;

fig4 =
 Table[Show[Plot[{Evaluate[IQ[t] /. solution[0.1, Parameters]]}, {t, t_start, t_end},
     AspectRatio → 0.4, ImageSize → 800, PlotRangePadding → None, Filling → Axis,
     PlotRange → {{tmin, tmax}, {ymin, ymax}}, AxesOrigin → {0, 0}, Frame →
      {{True, False}, {True, False}}, FrameStyle → Directive[Black, 17, Bold],
     PlotStyle → {{Thickness[0.01], RGBColor[28 / 255, 162 / 255, 0]}},
     FillingStyle -> Directive[Opacity[0.125]],
     FrameLabel → {{"Número de casos", None}, {None, None}},
     PlotLegends → Placed[{Table[Style[Row[{label}], Black, 17, "Text", Bold],
         {label, {"Redução de contactos em 90%"}}]},
       {Scaled[{0, 0.75}], {-1.25, 0.6}}], PlotLabel →
      Style["Evolução diária do número de casos confirmados ativos",
       17, Black, Bold],
     FrameTicks → {{Automatic, None}, {{{1 / 365, "02/03"},
         {PeakTiming90 / 365, Label90}, {PeakTimingBaseline / 365, LabelBaseline},
         {Emergencia, "18/03"}, {PeakTiming75 / 365, Label75}}, None}}],
    ListPlot[{DataBefore, DataAfter}],
    Graphics[{Black, Dashed, Thick,
      Line[{{Emergencia, ymin}, {Emergencia, ymax}}]}],
    Graphics[{RGBColor[28 / 255, 162 / 255, 0],
      Line[{{PeakTiming90 / 365, ymin}, {PeakTiming90 / 365, Peak90}}]}],
    Graphics[{RGBColor[28 / 255, 162 / 255, 0], Line[{{tmin, Peak90},
        {PeakTiming90 / 365, Peak90}}]}]], {i, 1, Length[vars]}][[1]]
```

```
Export[StringJoin[
    "//Users//LynxGAV//Documents//Work//CoronaPortugal//Figures//Figure1",
    ".pdf"], fig1];
Export[StringJoin[
    "//Users//LynxGAV//Documents//Work//CoronaPortugal//Figures//Figure2",
    ".pdf"], fig2];
Export[StringJoin[
    "//Users//LynxGAV//Documents//Work//CoronaPortugal//Figures//Figure3",
    ".pdf"], fig3];
Export[StringJoin[
    "//Users//LynxGAV//Documents//Work//CoronaPortugal//Figures//Figure4",
    ".pdf"], fig4];

ymax = Peak85 1.05;
ymin = -250;
tmax = 140 / 365;
tmin = t_start - 4 / 365;

fig5 =
 Table[Show[Plot[{Evaluate[IQ[t] /. solution[0.15, Parameters]]}, {t, t_start, t_end},
     AspectRatio → 0.4, ImageSize → 800, PlotRangePadding → None,
     Filling → Axis, PlotRange → {{tmin, tmax}, {ymin, ymax}},
     AxesOrigin → {0, 0}, Frame → {{True, False}, {True, False}},
     FrameStyle → Directive[Black, 17, Bold], PlotStyle →
      {{Thickness[0.01], Purple}}, FillingStyle -> Directive[Opacity[0.125]],
     FrameLabel → {{"Número de casos", None}, {None, None}},
     PlotLegends → Placed[{Table[Style[Row[{label}], Black, 17, "Text", Bold],
         {label, {"Redução de contactos em 85%"}}]},
       {Scaled[{0, 0.5}], {-1.49, 0.6}}], PlotLabel →
      Style["Evolução diária do número de casos confirmados ativos",
       17, Black, Bold],
     FrameTicks → {{Automatic, None}, {{{1 / 365, "02/03"},
         {PeakTiming85 / 365, Label85}, {PeakTiming90 / 365, Label90},
         {Emergencia, "18/03"}, {PeakTiming75 / 365, Label75}}, None}}],
    ListPlot[{DataBefore, DataAfter}],
    Graphics[{Black, Dashed, Thick,
      Line[{{Emergencia, ymin}, {Emergencia, ymax}}]}],
    Graphics[{Purple, Line[{{PeakTiming85 / 365, ymin},
       {PeakTiming85 / 365, Peak85}}]}],
    Graphics[{Purple, Line[{{tmin, Peak85}, {PeakTiming85 / 365, Peak85}}]}]],
   {i, 1, Length[vars]}][[1]]

Export[StringJoin[
    "//Users//LynxGAV//Documents//Work//CoronaPortugal//Figures//Figure5",
    ".pdf"], fig5];
```

Out[3755]=



Out[3760]=



Out[3765]=



Out[3770]=



Out[3779]=


Evolução diária do número de casos confirmados a
Número de casos
12 000
10 000
8000
6000
4000
2000
0
02/03
18/03
11/04
01/05
Redução d